STREAMING FREQUENCY-DOMAIN DAFX IN CSOUND 5

By V. Lazzarini, J. Timoney, T. Lysaght, NUI Maynooth, Ireland;

This work discusses the implementation of frequency domain digital audio effects using the Csound 5 music programming language, with its streaming frequency-domain signal (fsig) framework. It introduces the framework and its related unit generators. It describes in detail the different types of spectral DAFx made possible by these new opcodes.

The FSIG Framework

• Streaming frequency-domain signals are defined by the Csound fsig type.

• Such signals are processed at the rate of generation of new spectral frames.

• Fsigs currently support two types of data format: (1) Amplitude and Frequency (or phase) bin data, and (2) Partial Track data.

• Fsigs are self-describing, containing information about DFT length, hopsize, window size, window type and data format.

Spectral Analysis

Phase Vocoder:

```
asig inch     1
fs1  pvsanal  asig, 1024, 256, 1024, 1
```

Produces amplitude/frequency bin frame data using the standard PV analysis.

Instantaneous Frequency Distribution:

```
asig     inch    1
fs1,fs2  pvsifd  asig, 1024, 256, 1
```

Produces amplitude/frequency and amplitude/phase bin frame data using the IFD analysis.

Partial Track Analysis:

```
asig     inch    1
fs1,fs2  pvsifd  asig, 1024, 256, 1
ftrk  partials fs1, fs2, 0.003, 1,3,500
```

Produces partial track data from amplitude/frequency and (optionally) amplitude/phase bin frame data.

Spectral Processing

Amplitude Transformations

• Filtering

```
aphs phasor 1
afil oscili 2, kpks/2, 1
     tabw   abs(afil), aphs, 3, 1
ffil trfilter ftrk, 1, 3
```

• Noise Reduction/"Stencilling"

```
asig   diskin2  "input.wav", 1,0,1
ifn    ftgen 1,0, isiz/2,-43, "noise.pvx"
fsig   pvsanal  asig, isiz,isiz/4,isiz,iw
fclean pvstencil fsig, kattn, klvl,ifn
```

Frequency Transformations

• Frequency Scaling $f_{out}[np] = f_{in}[n]p$

```
fsig  pvsanal  asig, isiz,isiz/4,isiz,iw
ftps  pvscale  fsig, iscl,ikeepform,igain
```

• Frequency Shifting $f_{out}\left[n + \frac{s}{bw}\right] = f_{in}[n] + s$

```
ftrkd, ftrku trsplit ftrk, 1500
fshft        trshift ftrku, 150
ftmix        trmix   ftrkd, fshft
```



Formant Preservation under Frequency Transformations

Cross-synthesis and Other Transformations

• Morphing by interpolation of bin frames
• Channel vocoder-like amplitude substitution (`pvsvoc`)
• Partial track cross-synthesis (`trcross`)
• Reverse Panning (`pvsdemix`), loosely based on the ADRess algorithm
• Spectral Blurring and other smoothing effects
• Several other partial track effects

Resynthesis

• Overlap add inverse Phase Vocoder

• Bin frame Additive Synthesis

• Partial track Additive Synthesis
  • Linear
  • Cubic Phase

Conclusion

• The fsig framework in Csound5 and its spectral opcodes provide a comprehensive, flexible and intuitive way to build frequency-domain effects computer instruments.

• New opcodes will be continuously added to the existing set.

• Work on the sliding DFT analysis/resynthesis method by ffitch and Dobson will eventually be incorporated into the system.

Acknowledgements: The authors would like to acknowledge the work of Richard Dobson in the development of the fsig framework and opcodes for Csound 4.13. It is also important to mention the contribution of John ffitch and Istvan Varga, among others, to the development of the Csound 5 infra-structure.

```
instr 1
asrc   inch       1
aext   inch       2
andx   phasor     1
afln   oscili     10*gkfdf, gkfra, 1
       tablew     afln, andx, 3, 1
afil   oscili     10*gkfde, gkffr, 1
aran   randi      1, gkffr*2/gihbw
afil   =          aran*afil
anoi   rand       gknsg*0dbfs
aosc   oscili     gkosg*0dbfs, gkosp, 2
fsrc   pvsanal    asrc,1024,256,1024,1
fext   pvsanal    aext*gkexg,1024,256,1024,1
fosc   pvsanal    aosc,1024,256,1024,1
fnoi   pvsanal    anoi,1024,256,1024,1
ffil   pvsanal    afil,1024,256,1024,1
fscl   pvscale    fsrc, gkrat
fshf   pvshift    fsrc, gkshf
ffr    pvsmix     fscl, fshf
fexc1  pvsmix     fext, fnoi
fexc2  pvsmix     fosc, fexc1
fvoc   pvsvoc     fsrc, fexc2, 1, gknsg+gkosg+gkexg
flan   pvsmaska   fsrc, 3, gkfdf
ffs    pvsfilter  flan, ffil, gkfde
fblur  pvsblur    ffs, gkblt, 1
aout1  pvsynth    fvoc
aout2  pvsynth    fblur
aout3  pvsynth    ffr
amix   = aout1+aout2*(gkfdf+gkblt+gkfde)+ aout3*gkfdp
       outs       amix,amix
endin
```



A Csound Spectral DaFX Instrument